#No Riding Kill
execute as @e[tag=util.noriding_kill] unless function util:if/is_riding run kill @s
execute as @e[tag=util.noriding_kill.void] unless function util:if/is_riding run function util:void_kill

#No Passenger Kill
execute as @e[tag=util.nopassenger_kill] unless function util:if/have_passenger run kill @s
execute as @e[tag=util.nopassenger_kill.void] unless function util:if/have_passenger run function util:void_kill

#垃圾回收
kill @e[tag=util.trash]
tag @e[tag=son_entity] remove son_entity

#is c or p
tag @a[gamemode=creative] add is_c_or_p
tag @a[gamemode=spectator] add is_c_or_p
tag @a[gamemode=!creative,gamemode=!spectator] remove is_c_or_p

#scoreboards
    #generic_item_cd
    scoreboard players remove @a[scores={util.generic_item_cd=-10000..}] util.generic_item_cd 1
    scoreboard players reset @a[scores={util.generic_item_cd=..0}] util.generic_item_cd

#ItemEntityTpToOwner
execute as @e[tag=util.ItemEntityTPOwner] run function util:item_entity_tp_owner/tp with entity @s

#NoEntityOwnerKill
execute as @e[type=!marker,tag=util.NoEntityOwnerKill] run function util:entity_owner/no_entity_owner_kill with entity @s Passengers[{Tags:["util.entity_owner.marker"]}].data
execute as @e[type=marker,tag=util.NoEntityOwnerKill] run function util:entity_owner/no_entity_owner_kill with entity @s data